

PROHIBITION

```

10 DE=40 : REM POLICE !
20 CLS : REM EFFACE ECRAN
30 VM=900 : REM VAL DU MARCHE
40 P$(1)="RCC" : P$(2)="CO"
50 P$(3)="JO" : P$(4)="DAN"
60 DM = VM + INT(RND(1)*100)
70 AR=2000 : REM ARGENT JOUEUR
80 DM=VM+INT(RND(1)*100)

90 REM---<SORTIE DE TOLE !>-----
100 PRINT"VOUS SORTEZ DE TOLE !"
110 PRINT"- TON NOM ?"
120 INPUT "- ";N$
130 PRINT"- AH OUI, "N$
140 PRINT"- TIENS, V'LA "AR" $, "
150 PRINT" POUR TE REFAIRE, "N$
160 REM---<GARDE DU CORPS !>-----
170 PRINT"- TU VEUX UN GARDE DU CORPS (O/N) ?"
180 INPUT "- ";R$
190 IF R$<>"O" THEN 350
200 FOR I = 1 TO 4
210 N1=INT(RND(1)*4)+1:N2=INT(RND(1)*4)+1
220 GC$=P$(N1)+P$(N2)
230 PRINT"- JE TE PROPOSE "GC$
240 PG=INT(RND(1)*500)+200
250 PRINT"- IL PREND "PG " $ PAR JOUR"
260 PR=INT(RND(1)*50)+30
270 PRINT" IL TIRE BIEN : "PR"% DE REUSSITE"
280 PRINT"- C'EST OK OU UN AUTRE (O/N) ?"
290 INPUT "- ";R$
300 IF R$<>"O" THEN 320
310 PRINT"- OK , VA POUR "GC$ :GE=1 : GOTO 350
320 NEXT I
330 PRINT"- C'ETAIT LE DERNIER ! DE SOLE"

340 REM---<LA GNOLE>-----
350 PRINT : PRINT
360 PRINT"- HEY "N$", ON A "ST" LITRES EN STOCK "
370 PRINT" ET "AR" $ EN CAISSE !"
380 IF ST>300 THEN GOSUB 1450
390 PRINT:PRINT:PRINT"LE DEALER : " : PRINT
400 PRINT : PRINT"- HELLO ! T'EN VEUX COMBIEN ?"
410 INPUT "- ";NL
420 IF NL=0 THEN PRINT"- BON, TANT PIS" : GOTO 620
430 PA=NL^(-0.25): PA=INT(PA*40)
440 PRINT"- OUI..CA TE FAIT DU "PA" $/LITRE"
450 HT=PA*NL
460 PRINT" ET AU TOTAL "HT" $"
470 IF HT<=AR THEN 540
480 REM ---SPRGM USURIER---

```

```

490 IF PU=1 THEN PRINT"- T'AS PAS A SSEZ" : GOTO 400
500 GOSUB 1050
510 IF GE=0 THEN 170
520 GOTO 400
530 REM-----
540 PRINT"- C'EST OK ?"
550 INPUT "- ";R$
560 IF R$<>"O" THEN PRINT"- T'EN VEUX PLUS ?" : GOTO 390
570 AR=AR-HT : ST=ST+NL
580 PRINT"- OK "N$ ! TOPE LA !"
590 REM
600 REM
610 REM ----<LA VENTE !>-----
620 PRINT:PRINT"- HE DIS "N$;" ON M ET LE LITRE A COMBIEN ?"
630 INPUT "- ";PV(1)
640 PRINT"- OK !"
650 REM---<CALCUL CONCURRENCE>---
660 FOR I = 2 TO 5
670 PV(I) = 8 + INT(RND(1)*6)
680 S=S+PV(I)
690 NEXT I
700 MY=(S+PV(1))/5
710 PRINT"CONCURRENT "I" : " MY
720 IF MY>11 THEN DM=DM-INT((RND(1)*50)+150)
730 IF MY<10 THEN DM=DM+INT((RND(1)*50)+150)
740 REM ---<PARTAGE DU MARCHE>---
750 FOR I = 1 TO 5
760 D(I) = MY- PV(I)
770 PM(I)=INT( DM/5 + (D(I)*50))
780 IF PM(I)<=0 THEN PM(I)=0
790 PT=PT+PM(I)
800 NEXT I
810 REM---<RESULTATS>-----*
820 PRINT"DEMANDE EN VILLE : "PT" LITRES"
830 PRINT" ON AVAIT "ST" LITRES"
840 PRINT"- ON NOUS A DEMANDE "PM(1) "LITRES"
850 IF PM(1)>ST THEN 920
860 REM--STOCK SUFFISANT
870 ST=ST-PM(1)
880 CA = PM(1)* PV(1)
890 AR=AR+CA
900 GOTO 940
910 REM--STOCK INSUFFISANT--
920 CA=ST*PV(1)
930 ST=0 : AR=AR+CA
940 S=0 : PT=0
950 PRINT"- ON A GAGNE : "CA" $"
960 PRINT" ON A EN CAISSE : "AR" $"
970 IF GE=1 THEN AR=AR-PG
980 IF AR<0 THEN AR=0 : PRINT"- CIA O MINABLE !" : PRINT:PRINT"VOTRE GARDE DU CORPS EST PARTI !" : PRINT: GE=0
990 IF AR<=0 AND ST<=0 THEN PRINT"..PLUS UN RADIS ! ELIMINE " :END
1000 IF GE=1 THEN PRINT"- MOINS MON SALAIRE, "AR" $"
1010 PRINT" AU TOTAL : "AR " $"
1020 IF PU=0 THEN 1040
1030 US=US+1 : IF US=4 THEN US=0 : GOTO 1150
1040 GOTO 360
1050 REM---<L'USURIER>-----
1060 PRINT"- T'AS PAS ASSEZ" : PRINT" MAIS JE PEUX TE PRETER"
1070 NJ = INT(RND(1)*6)+1
1080 PRINT" TU RENDS LE DOUBLE DANS "NJ" JOURS"

```

```

1090 PRINT" OK ?"
1100 INPUT "- "; R$
1110 IF R$<>"O" THEN RETURN
1120 PRINT"- OK VOILA 2000 $"
1130 AR=AR+2000 : PU=1 : RETURN
1140 REM---RETOUR DE L'USURIER---
1150 PRINT:PRINT"-...BONJOUR ! "N$
1160 PRINT"- TU ME DOIS 4000 $"
1170 PRINT: PRINT"CHOIX : "
1180 PRINT"1. VOUS PAYEZ"
1190 PRINT"2. VOUS COUREZ"
1200 PRINT"3. VOUS TIREZ"
1210 IF GE=1 THEN PRINT"4. 'A TOI "GC$"" !"
1220 INPUT "- "; R
1230 ON R GOTO 1250,1280,1310,1400
1240 GOTO 1220
1250 REM --VOUS PAYEZ--
1260 IF AR<4000 THEN PRINT"- ON NE PEUT PAS "N$ : GOTO 1190
1270 AR=AR-4000 : GOTO 360
1280 REM --VOUS COUREZ--
1290 F=INT(RND(1)*100)
1300 IF F<60 THEN PRINT"- QUELLE COURSE ! OUF" : US=2 : GOTO 360
1310 REM--VOUS TIREZ--
1320 T=INT(RND(1)*100)
1330 IF T<20 THEN PRINT"- L'USURIER EST MORT !" : PU=0 : GOTO 360
1340 PRINT"- RATE ! IL TIRE A SON TOUR"
1350 U=INT(RND(1)*100)
1360 IF U<40 THEN PRINT"- VOUS ETES MORT !" : END
1370 BA=BA+1
1380 IF BA<6 THEN PRINT"VOUS RIPOSTEZ !" : GOTO 1320
1390 PRINT"VOUS RECHARGEZ " : GOTO 1180
1400 REM--GARDE DU CORPS --
1410 IF GE=0 THEN PRINT"VOUS REVEZ DU QUOI !" : GOTO 1110
1420 X=INT(RND(1)*100)
1430 IF X<PR THEN PRINT"- JE L'AI E U PATRON !" : PU=0 : GOTO 360
1440 PRINT"- ARGGGH PAAAATRON" : GE=0 : GOTO 1180
1450 REM---<LA POLICE>---
1460 PRINT: PRINT"- POLICE ! LES MAINS AU MUR"
1470 X= INT(RND(1)*100)
1480 IF X<DE THEN DE=DE+5 : EN=1
1490 PRINT"- FOUILLEZ MOI TOUT CA !"
1500 FOR I = 1 TO 500 : NEXT I
1510 IF EN=0 THEN PRINT"- ON A RIEN TROUVE CHEF" : PRINT"- BON BON...A BIENTOT !" : RETURN
1520 PRINT"- TIENS TIENS ! "ST" LITRES DE GNOLE !"
1530 IF AR<=0 THEN PRINT"- ALLEZ, A U GNOUF !" : PRINT: PRINT"VOUS RETOURNEZ EN PRISON !" : PRINT"...6 MOIS PLUS TARD..." : GOTO 30
1540 PRINT"- T'ES FAIT...A MOINS QU'E..."
1550 PRINT" COMBIEN POUR LES OEUVRES..?"
1560 INPUT "- ";R
1570 IF R<(AR/2) THEN 1590
1580 PRINT"- BON, CA IRA POUR CETTE FOIS" :AR=AR-R : RETURN
1590 PRINT"- HE HE ! TU RIS. "N$"
1600 PRINT"- ALLEZ, PRENEZ TOUT !"
1610 ST=0 : RETURN

```